Costruzione delle funzioni di reazione nel modello di Cournot

Gianmaria Martini

Introduzione

Questo documento mostra come costruire la funzione di reazione nel modello di Cournot e, soprattutto, il suo significato economico.

Si parta da una funzione di domanda lineare, in cui l'offerta è data da due imprese, 1 e 2.

```
> p:=a-b*(y1+y2);
```

$$p := a - b\,(y1 + y2)$$

La funzione del profitto dell'impresa 1 è data da, ipotizzando una funzione di costo a rendimenti costanti *cyi*

```
> pi1:=p*y1-c*y1;
```

$$\pi 1 := (a - b\,(y1 + y2))\,y1 - c\,y1$$

Possiamo assumere dei dati al problema, che poi e' possibile cambiare a piacimento

```
> data:=(a=100, b=2, c=4);
```

$$data := a = 100,\ b = 2,\ c = 4$$

```
> subs(data, pi1);
```

$$(100 - 2\,y1 - 2\,y2)\,y1 - 4\,y1$$

```
> pi1:=subs(data,pi1);
```

$$\pi 1 := (100 - 2\,y1 - 2\,y2)\,y1 - 4\,y1$$

Rispetto alla suindicata funzione del profitto dell'impresa 1, possiamo individuare due livelli di $y1$ cui corrisponde un profitto nullo: $y1 = 0$ e il seguente (definito come h per non confondere il sistema)

```
> h:=(a-c-b*y2)/b;
```

$$h := \frac{a - c - b\,y2}{b}$$

```
> h:=subs(data,h);
```

$$h := 48 - y2$$

Spiegazione grafica del concetto di funzione di reazione

```
> with(plots):
Warning, the name changecoords has been redefined
```

Possiamo rappresentare graficamente la funzione del profitto al variare di *y2*.

```
> animate(pi1, y1=0..96/2, y2=0..50, color=blue, thickness=3,
  title="Funzione del profitto in Cournot");
```



Allo stesso modo rappresentiamo in un grafico a 3 dimensioni la funzione del profitto. Cio' che interessa l'impresa e' il punto di massimo.

```
> plot3d(pi1, y1=0..h, y2=0..96/2, title="Funzione del profitto
  in Cournot a 3 dimensioni");
```

Funzione del profitto in Cournot a 3 dimensioni




```
> restart;
```

Siamo adesso in grado di capire cosa rappresenta la funzione di reazione: e' la proiezione in un grafico a 2 dimensioni del massimo della funzione a 3 dimensioni, per ogni dato livello di *y2*.

Prendiamo allora una generica funzione di domanda lineare

```
> p:=a-b*(y1+y2);
```

$$p := a - b\,(y1 + y2)$$

scriviamo la funzione del profitto

```
> pi1:=p*y1-c*y1;
```

$$\pi 1 := (a - b\,(y1 + y2))\,y1 - c\,y1$$

determiniamo la condizione del primo ordine per il massimo profitto

```
> cpoy1:=diff(pi1,y1);
```

$$cpoy1 := -b\,y1 + a - b\,(y1 + y2) - c$$

risolviamola per *y1*

```
> rea1:=solve(cpoy1,y1);
```

$$rea1 := -\frac{1}{2}\,\frac{c - a + b\,y2}{b}$$

otteniamo la funzione di reazione dell'impresa 1. Essa e' funzione di *y2*. Quindi rappresenta la *best reply*. Se sostituiamo i dati otteniamo

```
> data:=(a=100, b=2, c=4);
```

$$data := a = 100,\ b = 2,\ c = 4$$

```
> rea1:=subs(data,rea1);
```

$$rea1 := 24 - \frac{1}{2}\,y2$$

possiamo rappresentarla graficamente in 2 dimensioni.

```
> plot(rea1, y2=0..48, thickness=3, color=blue, title="Funzione di reazione impresa 1");
```




```
> restart;
```

Statica comparata

Possiamo a questo punto rappresentare come la funzione di reazione varia al variare di qualche parametro (a, b, c)

```
> p:=a-b*(y1+y2);
```

$$p := a - b\,(y1 + y2)$$

```
> pi1:=p*y1-c*y1;
```

$$\pi 1 := (a - b\,(y1 + y2))\,y1 - c\,y1$$

```
> cpoy1:=diff(pi1,y1);
```

$$cpoy1 := -b\,y1 + a - b\,(y1 + y2) - c$$

```
> rea1:=solve(cpoy1,y1);
```

$$rea1 := -\frac{1}{2}\,\frac{c - a + b\,y2}{b}$$

```
> data:=(a=100, c=4);
```

$$data := a = 100,\ c = 4$$

```
> rea1:=subs(data,rea1);
```

$$rea1 := -\frac{1}{2}\,\frac{-96 + b\,y2}{b}$$

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> animate(rea1, y2=0..100, b=1..5, color=blue,
  thickness=3,title="Variazione della pendenza della domanda e
  funzione di reazione");
```



Equilibrio di Cournot Nash

Determiniamo anche la funzione di reazione dell'impresa 2

```
> pi2:=p*y2-c*y2;
```

$$\pi 2 := (a - b\,(y1 + y2))\,y2 - c\,y2$$

```
> cpoy2:=diff(pi2,y2);
```

$$cpoy2 := -b\,y2 + a - b\,(y1 + y2) - c$$

```
> rea2:=solve(cpoy2,y2);
```

$$rea2 := -\frac{1}{2}\,\frac{c - a + b\,y1}{b}$$

```
> rea2:=subs(data,rea2);
```

$$rea2 := -\frac{1}{2}\,\frac{-96 + b\,y1}{b}$$

```
> b:=2;
```

$$b := 2$$

```
> rea2;
```

$$24 - \frac{1}{2}\,y1$$

invertiamo la funzione di reazione dell'impresa 1 per rappresentarla nello stesso grafico della funzione di reazione dell'impresa 2

```
> rea11:=(a-c)/b-2*y1;
```

$$rea11 := \frac{1}{2}\,a - \frac{1}{2}\,c - 2\,y1$$

```
> rea11:=subs(data,rea11);
```

$$rea11 := 48 - 2\,y1$$

```
> rea111:=y1 -> piecewise(y1>=0 and y1<=24, rea11, 0);
```

$$rea111 := y1 \rightarrow \mathrm{piecewise}(0 \le y1 \textbf{ and } y1 \le 24, rea11, 0)$$

```
> rea111(y1);
```

$$\begin{cases} 48 - 2\,y1 & -y1 \le 0 \textbf{ and } y1 - 24 \le 0 \\ 0 & otherwise \end{cases}$$

a questo punto possiamo rappresentarle insieme

```
> plot([rea111(y1),rea2], y1=0..48, thickness=3,
  color=[blue,red], title="Equilibrio di Cournot - Nash");
```



L'intersezione tra le due funzioni di reazione e' l'equilibrio di Nash.

```
> restart;
```

Statica comparata sull'equilibrio di Nash

E' anche possibile vedere l'effetto sull'equilibrio di una variazione di uno dei tre parametri (a,b,c).

```
> p:=a-b*(y1+y2);
```

$$p := a - b\,(y1 + y2)$$

```
> pi1:=p*y1-c*y1;
```

$$\pi 1 := (a - b\,(y1 + y2))\,y1 - c\,y1$$

```
> data:=(a=100, b=2);
```

$$data := a = 100,\ b = 2$$

```
> pi1:=subs(data,pi1);
```

$$\pi 1 := (100 - 2\,y1 - 2\,y2)\,y1 - c\,y1$$

```
> cpo1:=diff(pi1,y1);
```

$$cpo1 := -4\,y1 + 100 - 2\,y2 - c$$

```
> rea1:=solve(cpo1,y2);
```

$$rea1 := -2\,y1 + 50 - \frac{1}{2}\,c$$

```
> pi2:=subs(data,p*y2-c*y2);
```

$$\pi 2 := (100 - 2\,y1 - 2\,y2)\,y2 - c\,y2$$

```
> cpo2:=diff(pi2,y2);
```

$$cpo2 := -4\,y2 + 100 - 2\,y1 - c$$

```
> rea2:=solve(cpo2,y2);
```

$$rea2 := 25 - \frac{1}{2}\,y1 - \frac{1}{4}\,c$$

```
> h:=25-(1/4)*c;
```

$$h := 25 - \frac{1}{4}\,c$$

```
> k:=50-(1/2)*c;
```

$$k := 50 - \frac{1}{2}\,c$$

```
> rea11:=y1 -> piecewise(y1>=0 and y1<=h, rea1, 0);
```

$$rea11 := y1 \to \mathrm{piecewise}(0 \le y1 \textbf{ and } y1 \le h, rea1, 0)$$

```
> rea11(y1);
```

$$\begin{cases} -2\,y1 + 50 - \frac{1}{2}\,c & -y1 \le 0 \textbf{ and } y1 - 25 + \frac{1}{4}\,c \le 0 \\ 0 & \mathit{otherwise} \end{cases}$$

```
> rea22:=y1 -> piecewise(y1>=0 and y1<=k, rea2,0);
```

$$rea22 := y1 \to \mathrm{piecewise}(0 \le y1 \textbf{ and } y1 \le k, rea2, 0)$$

```
> rea22(y1);
```

$$\begin{cases} 25 - \frac{1}{2}\,y1 - \frac{1}{4}\,c & -y1 \le 0 \textbf{ and } y1 - 50 + \frac{1}{2}\,c \le 0 \\ 0 & \mathit{otherwise} \end{cases}$$

```
> with(plots):
Warning, the name changecoords has been redefined

> a:=animate(rea11(y1),y1=0..50,c=4..30, color=blue,
  thickness=3, title="Incremento dei costi e equilibrio di
  Cournot"):
> b:=animate(rea22(y1),y1=0..50,c=4..30, color=red,
  thickness=3):
> display(a,b);
```



>

>

ad esempio, se crescono i costi (*c* varia da 4 a 30), l'equilibrio si sposta verso l'origine. Quindi le imprese producono meno, e scaricano l'aumento dei costi sul prezzo.